
Sumos in competition
Sumobots in competition
Your sumobots in competition
Rules

· Match Start: The match begins when two competitors agree to battle and its their turn on a ring.

· Match End: The match is over when either 3 minutes has elapsed or 3 bouts have concluded.

· Match Scoring: After the match, the robot with the most bout points is the winner. Match points are equivalent to the bout points:

· 1 point for a win

· .5 point for a draw

· 0 points for a loss

· Match Ties: Ties are not allowed; all matches must end with a winner. If the match clock runs out during the final bout, the robots are allowed to play on until a winner is found.

· Match Start: The robots are placed on the sumo ring according to the positioning rules. The builders will indicate their readiness and on the time keeper's signal, will start their robots. Each robot should have a 5 second delay after the builder starts the robot before the robot starts to move.

· Match End: A match ends when...

· ...the first robot touches the area outside the ring.

· ...the robots are deadlocked for 30 seconds.

· ...both builders agree on a draw.

· ...the match time expires.

Source of all Information: The SumoBot Manual
Let’s Begin at the beginning

Make it move: Servos

The program:

' {$STAMP BS2}

' {$PBASIC 2.5}

' -----[ I/O Definitions ]-----------------------------------------------------
LMotor CON 13                                ' left servo motor

RMotor CON 12                                ' right servo motor

' -----[ Constants ]-----------------------------------------------------------
LFwdFast CON 1000                            ' left motor forward; fast

LFwdSlow CON 800                             ' left motor forward; slow

LStop CON 750                                ' left motor stop

LRevSlow CON 700                             ' left motor reverse; slow

LRevFast CON 500                             ' left motor reverse; fast

RFwdFast CON 500                             ' right motor forward; fast

RFwdSlow CON 700                             ' right motor forward; slow

RStop CON 750                                ' right motor stop

RRevSlow CON 800                             ' right motor reverse; slow

RRevFast CON 1000                            ' right motor reverse; fast

' -----[ Variables ]-----------------------------------------------------------
pulses VAR Byte                              ' counter for motor control

' -----[ Initialization ]------------------------------------------------------
Start_Delay:

PAUSE 2000                                   ' time to disconnect cable

' -----[ Main Code ]-----------------------------------------------------------
Main:

   FOR pulses = 1 TO 125                     ' forward slow - about 12 inches

      PULSOUT LMotor, LFwdSlow

      PULSOUT RMotor, RFwdSlow

      PAUSE 20

   NEXT

   FOR pulses = 1 TO 110                     ' pivot turn - left wheel (ccw)

      PULSOUT LMotor, LStop

      PULSOUT RMotor, RFwdSlow

      PAUSE 20

   NEXT

   FOR pulses = 1 TO 50                      ' forward fast

      PULSOUT LMotor, LFwdFast

      PULSOUT RMotor, RFwdFast

      PAUSE 20

   NEXT

   FOR pulses = 1 TO 30                      ' spin turn - clockwise

      PULSOUT LMotor, LFwdFast

      PULSOUT RMotor, RRevFast

      PAUSE 20

   NEXT

Hold_Position:

   PULSOUT LMotor, LStop

   PULSOUT RMotor, RStop

   PAUSE 20

   GOTO Hold_Position

Line Sensors

The Program 1:
' {$STAMP BS2}

' {$PBASIC 2.5}

' -----[ I/O Definitions ]-----------------------------------------------------
LLineSnsrPwr CON 10                                  ' left line sensor power

LLineSnsrIn CON 9                                    ' left line sensor input

RLineSnsrPwr CON 7                                   ' right line sensor power

RLineSnsrIn CON 8                                    ' right line sensor input

' -----[ Variables ]-----------------------------------------------------------
leftSense VAR Word                                   ' left sensor raw reading

rightSense VAR Word                                  ' right sensor raw reading

' -----[ Main Code ]-----------------------------------------------------------
Read_Left:

   HIGH LLineSnsrPwr                                 ' activate sensor

   HIGH LLineSnsrIn                                  ' discharge QTI cap

   PAUSE 1

   RCTIME LLineSnsrIn, 1, leftSense                  ' read sensor value

   LOW LLineSnsrPwr                                  ' deactivate sensor

Read_Right:

   HIGH RLineSnsrPwr                                 ' activate sensor

   HIGH RLineSnsrIn                                  ' discharge QTI cap

   PAUSE 1

   RCTIME RLineSnsrIn, 1, rightSense                 ' read sensor value

   LOW RLineSnsrPwr                                  ' deactivate sensor

Display:

   DEBUG Home

   DEBUG "Left ", TAB, "Right", CR

   DEBUG "-----", TAB, "-----", CR

   DEBUG DEC leftSense, ClrEOL, TAB, DEC rightSense, ClrEOL

   PAUSE 100

GOTO Read_Left

The Program 2:

' {$STAMP BS2}

' {$PBASIC 2.5}

' -----[ I/O Definitions ]-----------------------------------------------------
LMotor CON 13                                                 ' left servo motor

RMotor CON 12                                                 ' right servo motor

LLineSnsrPwr CON 10                                           ' left line sensor power

LLineSnsrIn CON 9                                             ' left line sensor input

RLineSnsrPwr CON 7                                            ' right line sensor power

RLineSnsrIn CON 8                                             ' right line sensor input

StartLED CON 0                                                ' display start delay

' -----[ Constants ]------------------------------------------------------------

LFwdFast CON 1000                                             ' left motor forward; fast

LFwdSlow CON 800                                              ' left motor forward; slow

LStop CON 750                                                 ' left motor stop

LRevSlow CON 700                                              ' left motor reverse; slow

LRevFast CON 500                                              ' left motor reverse; fast

RFwdFast CON 500                                              ' right motor forward; fast

RFwdSlow CON 700                                              ' right motor forward; slow

RStop CON 750                                                 ' right motor stop

RRevSlow CON 800                                              ' right motor reverse; slow

RRevFast CON 1000                                             ' right motor reverse; fast

ClrEOL CON 11                                                 ' clear to end of line (DEBUG)

' -----[ Variables ]------------------------------------------------------------

leftSense VAR Word                                            ' left sensor raw reading

rightSense VAR Word                                           ' right sensor raw reading

lineBits VAR Nib                                              ' decoded sensors value

lineLeft VAR lineBits.Bit1

lineRight VAR lineBits.Bit0

pulses VAR Byte                                               ' counter for motor control

temp VAR Byte

RunStatus DATA $00                                            ' run status
leftThresh VAR word

rightThresh VAR word

' -----[ Initialization ]-------------------------------------------------------

Run_Check:                                                    ' user Reset button as On-Off

   READ RunStatus, temp                                       ' read current status

   temp = ~temp                                               ' invert status

   WRITE RunStatus, temp                                      ' save status for next reset

   IF (temp = 0) THEN Start_Delay                             ' run now?

   END                                                        ' -- no ... next time
Set_Threshold:                                                ' set QTI black threshold

   GOSUB Read_Line_Sensors

   leftThresh = lineLeft / 2

   rightThresh = lineRight / 2

Start_Delay:                                                  ' mandatory 5 second delay

   HIGH StartLED                                              ' show active

   PAUSE 5000

   INPUT StartLED                                             ' LED off

' -----[ Main Code ]------------------------------------------------------------

Main:

   GOSUB Read_Line_Sensors

' Sumo movement

   BRANCH lineBits, [Go_Fwd, Spin_Left, Spin_Right, About_Face]

Go_Fwd:

   PULSOUT LMotor, LFwdFast

   PULSOUT RMotor, RFwdFast
GOTO Main

Spin_Left:

   FOR pulses = 1 TO 20

      PULSOUT LMotor, LRevFast

      PULSOUT RMotor, RFwdFast

      PAUSE 20

   NEXT

   GOTO Main

Spin_Right:

   FOR pulses = 1 TO 20

      PULSOUT LMotor, LFwdFast

      PULSOUT RMotor, RRevFast

      PAUSE 20

   NEXT

   GOTO Main

About_Face:

   FOR pulses = 1 TO 10                                       ' back up from edge

      PULSOUT LMotor, LRevFast

      PULSOUT RMotor, RRevFast

      PAUSE 20

   NEXT

   FOR pulses = 1 TO 30                                       ' turn around

      PULSOUT LMotor, LFwdFast

      PULSOUT RMotor, RRevFast

      PAUSE 20

   NEXT

   GOTO Main

' -----[ Subroutines ]----------------------------------------------------------

Read_Line_Sensors:

   HIGH LLineSnsrPwr                                          ' activate sensors

   HIGH RLineSnsrPwr

   HIGH LLineSnsrIn                                           ' discharge caps

   HIGH RLineSnsrIn

   PAUSE 1

   RCTIME LLineSnsrIn, 1, leftSense                           ' read left sensor

   RCTIME RLineSnsrIn, 1, rightSense                          ' read right sensor

   LOW LLineSnsrPwr ' deactivate sensors

   LOW RLineSnsrPwr

' convert readings to bits

   LOOKDOWN leftSense, >=[leftThresh, 0], lineLeft            ' 0 = on black, 1 = on line

   LOOKDOWN rightSense, >=[rightThresh, 0], lineRight

RETURN
Infrared Object Detection
[image: image6.emf]
Program

' {$STAMP BS2}

' {$PBASIC 2.5}

' -----[ I/O Definitions ]-------------------------------------------------

LMotor PIN 13                        ' left servo motor

RMotor PIN 12                        ' right servo motor

LfIrOut PIN 4                        ' left IR LED output

LfIrIn PIN 11                        ' left IR sensor input

RtIrOut PIN 15                       ' right IR LED output

RtIrIn PIN 14                        ' right IR sensor input

' -----[ Constants ]-------------------------------------------------------

LFwdFast CON 1000                    ' left motor fwd; fast

LFwdSlow CON 800                     ' left motor fwd; slow

LStop CON 750                        ' left motor stop

LRevSlow CON 700                     ' left motor rev; slow

LRevFast CON 500                     ' left motor rev; fast

RFwdFast CON 500                     ' right motor fwd; fast

RFwdSlow CON 700                     ' right motor fwd; slow

RStop CON 750                        ' right motor stop

RRevSlow CON 800                     ' right motor rev; slow

RRevFast CON 1000                    ' right motor rev; fast

' -----[ Variables ]-------------------------------------------------------

irBits VAR Nib                       ' storage for IR target data

irLeft VAR irBits.BIT1

irRight VAR irBits.BIT0

lastIr VAR Nib                       ' info from last reading

pulses VAR Byte                      ' counter for motor control

' -----[ Initialization ]--------------------------------------------------

Reset:

   LOW LMotor                        ' initialize motor outputs

   LOW RMotor

' -----[ Program Code ]----------------------------------------------------
Main:

   GOSUB Read_IR_Sensors

   BRANCH irBits, [Scan, Follow_Right, Follow_Left, Hold]

Scan:

   BRANCH lastIR, [Move_Fwd, Scan_Right, Scan_Left, Move_Fwd]

Move_Fwd:

   DEBUG HOME, "Forward", CLREOL

   GOTO Main

Scan_Right:                             ' spin right, slow

   DEBUG HOME, "Scan Right", CLREOL

   PULSOUT LMotor, LFwdSlow

   PULSOUT RMotor, RRevSlow

   PAUSE 20

   GOTO Main

Scan_Left:                              ' spin left, slow

   DEBUG HOME, "Scan Left", CLREOL

   PULSOUT LMotor, LRevSlow

   PULSOUT RMotor, RFwdSlow

   PAUSE 20

   GOTO Main

Follow_Right:                            ' spin right, fast

   DEBUG HOME, "Follow Right", CLREOL

   PULSOUT LMotor, LFwdFast

   PULSOUT RMotor, RRevFast

   PAUSE 20

   lastIr = irBits                       ' save last direction found

   GOTO Main

Follow_Left:                             ' spin left, fast

   DEBUG HOME, "Follow Left", CLREOL

   PULSOUT LMotor, LRevFast

   PULSOUT RMotor, RFwdFast

   PAUSE 20

   lastIr = irBits

   GOTO Main

Hold:                                     ' on target

   DEBUG HOME, "On Target", CLREOL

   FOR pulses = 1 TO 3

      PULSOUT LMotor, LStop

      PULSOUT RMotor, RStop

      PAUSE 20

   NEXT

   lastIr = %00

   GOTO Main

END

' -----[ Subroutines ]-----------------------------------------------------

   Read_IR_Sensors:

   FREQOUT LfIrOut, 1, 38500              ' modulate left IR LED

   irLeft = ~LfIrIn                       ' read input (1 = target)

   FREQOUT RtIrOut, 1, 38500              ' modulate right IR LED

   irRight = ~RtIrIn                      ' read input (1 = target)

RETURN

Competition Code

Program

' {$STAMP BS2}

' {$PBASIC 2.5}

' -----[ I/O Definitions ]-------------------------------------------------

LMotor PIN 13                                ' left servo motor

RMotor PIN 12                                ' right servo motor

LLinePwr PIN 10                              ' left line sensor power

LLineIn PIN 9                                ' left line sensor input

RLinePwr PIN 7                               ' right line sensor power

RLineIn PIN 8                                ' right line sensor input

LfIrOut PIN 4                                ' left IR LED output

LfIrIn PIN 11                                ' left IR sensor input

RtIrOut PIN 15                               ' right IR LED output

RtIrIn PIN 14                                ' right IR sensor input

Speaker PIN 1                                ' piezo speaker

StartLED PIN 0                               ' display start delay

' -----[ Constants ]-------------------------------------------------------

LFwdFast CON 1000                            ' left motor fwd; fast

LFwdSlow CON 800                             ' left motor fwd; slow

LStop CON 750                                ' left motor stop

LRevSlow CON 700                             ' left motor rev; slow

LRevFast CON 500                             ' left motor rev; fast

RFwdFast CON 500                             ' right motor fwd; fast

RFwdSlow CON 700                             ' right motor fwd; slow

RStop CON 750                                ' right motor stop

RRevSlow CON 800                             ' right motor rev; slow

RRevFast CON 1000                            ' right motor rev; fast

' -----[ Variables ]-------------------------------------------------------

lLine VAR Word                               ' left sensor raw reading

rLine VAR Word                               ' right sensor raw reading

blackThresh VAR Word                         ' QTI black threshold

lineBits VAR Nib                             ' decoded sensors value

lbLeft VAR lineBits.BIT1

lbRight VAR lineBits.BIT0

irBits VAR Nib                               ' IR readings (l & r)

irLeft VAR irBits.BIT1

irRight VAR irBits.BIT0

lastIr VAR Nib                               ' info from last reading

pulses VAR Byte                              ' counter for motor control

temp VAR Byte

RunStatus DATA $00                           ' run status

' -----[ Initialization ]--------------------------------------------------

Reset:

READ RunStatus, temp                         ' read current status

temp = ~temp                                 ' invert status

WRITE RunStatus, temp                        ' save for next reset

IF (temp > 0) THEN END                       ' okay to run?

' SumoBot must be placed over black playing surface before this code runs.

Set_Threshold:                               ' set QTI black threshold

   GOSUB Read_Line_Sensors

   leftThresh = lLine / 2

   rightThresh = rLine / 2

   LOW LMotor                                ' make more pins outputs

   LOW RMotor

Start_Delay:                                 ' five second delay

   FOR temp = 1 TO 5

      HIGH StartLED                          ' show active

      PAUSE 900

      INPUT StartLED                         ' blink each second

      FREQOUT Speaker, 100, 2500, 3000       ' beep each second

   NEXT

   GOTO Lunge                                ' start aggressive!

' -----[ Program Code ]----------------------------------------------------

Main:

   GOSUB Read_Line_Sensors

' If not on the Shikiri line (border), continue to look for opponent,

' otherwise, spin back toward center and resume search

   BRANCH lineBits, [Search_For_Opponent, Spin_Left, Spin_Right, About_Face]

' --[ Border Avoidance ]--

Spin_Left:                                   ' right sensor was active

   FOR pulses = 1 TO 20

      PULSOUT LMotor, LRevFast

      PULSOUT RMotor, RFwdFast

      PAUSE 20

   NEXT

   lastIr = %00                              ' clear scan direction

   GOTO Lunge

Spin_Right:                                  ' left sensor was active

   FOR pulses = 1 TO 20

      PULSOUT LMotor, LFwdFast

      PULSOUT RMotor, RRevFast

      PAUSE 20

   NEXT

   lastIr = %00

   GOTO Lunge

About_Face:                                  ' both sensors on Shikiri

   FOR pulses = 1 TO 10                      ' back up from edge

      PULSOUT LMotor, LRevFast

      PULSOUT RMotor, RRevFast

      PAUSE 20

   NEXT

   FOR pulses = 1 TO 30                      ' turn around

      PULSOUT LMotor, LFwdFast

      PULSOUT RMotor, RRevFast

      PAUSE 20

   NEXT

   lastIr = %00

   GOTO Lunge

' --[ IR Processing ]--

Search_For_Opponent:

   GOSUB Read_IR_Sensors

' If opponent is not in view, scan last known direction. Turn toward

' opponent if seen by one "eye" -- if both, lunge forward

   BRANCH irBits, [Scan, Follow_Right, Follow_Left, Lunge]

Scan:

   BRANCH lastIR, [Move_Fwd, Scan_Right, Scan_Left]

Move_Fwd:

   GOSUB Creep_Forward

   GOTO Main

Scan_Right:                                   ' spin right, slow

   FOR pulses = 1 TO 5

       PULSOUT LMotor, LFwdSlow

       PULSOUT RMotor, RRevSlow

       PAUSE 20

   NEXT

   GOSUB Creep_Forward                        ' keep moving

   GOTO Main

Scan_Left:                                    ' spin left, slow

   FOR pulses = 1 TO 5

       PULSOUT LMotor, LRevSlow

       PULSOUT RMotor, RFwdSlow

       PAUSE 20

   NEXT

   GOSUB Creep_Forward

   GOTO Main

Follow_Right:                                 ' spin right, fast

   PULSOUT LMotor, LFwdFast

   PULSOUT RMotor, RRevSlow

   lastIR = irBits ' save last direction found

   GOTO Main

Follow_Left:                                  ' spin left, fast

   PULSOUT LMotor, LRevSlow

   PULSOUT RMotor, RFwdFast

   lastIR = irBits

   GOTO Main

Lunge:                                        ' locked on -- go get him!

   FOR pulses = 1 TO 25

      PULSOUT LMotor, LFwdFast

      PULSOUT RMotor, RFwdFast

      GOSUB Read_Line_Sensors

      IF (lineBits = %11) THEN Match_Over     ' in sight, we're on the line

    NEXT

    GOTO Main

' If SumoBot can see the opponent with both "eyes" and both QTIs are

' detecting the border, we must have pushed the opponent out.

Match_Over:

   FOR pulses = 1 TO 10                       ' stop motors

       PULSOUT LMotor, LStop

       PULSOUT RMotor, RStop

       PAUSE 20

   NEXT

   INPUT LMotor

   INPUT RMotor

   FOR temp = 1 TO 10                         ' make some noise

       HIGH StartLED

       FREQOUT Speaker, 100, 2500, 3000       ' beep

       INPUT StartLED                         ' blink LED

       PAUSE 100

   NEXT

   GOTO Reset                                 ' reset for next round

' -----[ Subroutines ]-----------------------------------------------------

Read_Line_Sensors:

   HIGH LLinePwr                              ' activate sensors

   HIGH RLinePwr

   HIGH LLineIn                               ' discharge caps

   HIGH RLineIn

   PAUSE 1

   RCTIME LLineIn, 1, lLine                   ' read left sensor

   RCTIME RLineIn, 1, rLine                   ' read right sensor

   LOW LLinePwr                               ' deactivate sensors

   LOW RLinePwr

' convert readings to bits: 0 = on black, 1 = on line
   LOOKDOWN leftSense, >=[leftThresh, 0], lineLeft           

   LOOKDOWN rightSense, >=[rightThresh, 0], lineRight

RETURN

Read_IR_Sensors:

   FREQOUT LfIrOut, 1, 38500                  ' modulate left IR LED

   irLeft = ~LfIrIn                           ' read input (1 = target)

   FREQOUT RtIrOut, 1, 38500                  ' modulate right IR LED

   irRight = ~RtIrIn                          ' read input (1 = target)

RETURN

Creep_Forward:

   FOR pulses = 1 TO 20

      PULSOUT LMotor, LFwdSlow

      PULSOUT RMotor, RFwdSlow

      PAUSE 20

   NEXT

RETURN

Code from Wayne Lowery

' {$STAMP BS2}

' {$PBASIC 2.5}

' Wayne Lowery

' m2n@hotmail.com

' Modified sumobot program for Parallax SumoBot board

'

' Variables and constents

'

' Line sensor

LLineSnsEmit    PIN     10

RLineSnsEmit    PIN     7

LLineSnsIn      PIN     9                      ' left line sensor input

RLineSnsIn      PIN     8                      ' right line sensor input

leftSense       VAR     Word                   ' left sensor raw reading

rightSense      VAR     Word                   ' right sensor raw reading

blackThreshL    VAR     Word                   ' black threshold setting

blackThreshR    VAR     Word                   ' black threshold setting

lineBits        VAR     Nib                    ' decoded sensors value

' IR sensor

LfIrOut         PIN     4                      ' left IR LED output

LfIrIn          PIN     11                     ' left IR sensor input

RtIrOut         PIN     15                     ' right IR LED output

RtIrIn          PIN     14                     ' right IR sensor input

irBits          VAR     Nib                    ' storage for IR target data

irBitsLast      VAR     Nib

' Servos

Rservo           PIN     12

Lservo           PIN     13

SStop            CON     750

fast             CON     50

slow             CON     20

temp            VAR     Byte

' Startup Status

Speaker PIN 1                                ' piezo speaker

StartLED PIN 0                               ' display start delay

RunStatus DATA $00                           ' run status

' ========================================================================

' Reset Button Start

Reset:

READ RunStatus, temp                         ' read current status

temp = ~temp                                 ' invert status

WRITE RunStatus, temp                        ' save for next reset

IF (temp > 0) THEN END                       ' okay to run?

' ========================================================================

' Initialize

'

GOSUB Read_Line_Sensors

blackThreshL = (3*leftSense / 6)

blackThreshR = (3*rightSense / 6)            'orig 2/3

'debugLoop:

'DO

'  GOSUB Read_Line_Sensors

'  GOSUB Read_IR_Sensors

'  GOSUB debugMsg

'  PAUSE 500

'LOOP

irBitsLast=%0011

Start_Delay:                                 ' five second delay

   FOR temp = 1 TO 5

      HIGH StartLED                          ' show active

      PAUSE 900

      INPUT StartLED                         ' blink each second

      FREQOUT Speaker, 100, 2500, 3000       ' beep each second

   NEXT

GOSUB forward

'

' Start main loop

'

main:

DO

  GOSUB Read_Line_Sensors

  IF (linebits <> %0011) THEN                        ' Line detected, act immediatly

    GOSUB backup

    IF linebits = %0001 THEN

      FOR temp = 1 TO 35

        GOSUB right

      NEXT

    ELSE

      FOR temp = 1 TO 35

        GOSUB left

      NEXT

    ENDIF

  ELSE

    GOSUB Read_IR_Sensors

    IF irBits = 0 THEN

      irBits=irBitsLast

    ELSE

      irBitsLast = irBits

    ENDIF

    ON irBits GOSUB forward, right, left, forward

  ENDIF

'    GOSUB debugMsg                                     ' Display debug messages, comment out for normal use

'    PAUSE 500                                          ' pause for debugging, comment out for normal use

LOOP

' ========================================================================

' Start subroutines

'

' Sensor updates

Read_Line_Sensors:

  HIGH LLineSnsEmit

  HIGH LLineSnsIn

  PAUSE 1

  RCTIME LLineSnsIn, 1, leftSense

  LOW LLineSnsEmit

  HIGH RLineSnsEmit

  HIGH RLineSnsIn

  PAUSE 1

  RCTIME RLineSnsIn, 1, rightSense

  LOW RLineSnsEmit

  ' convert readings to bits

  LOOKDOWN leftSense, >=[blackThreshL, 0], lineBits.BIT1

  LOOKDOWN rightSense, >=[blackThreshR, 0], lineBits.BIT0

  lineBits = ~lineBits                          ' 1 for line, 0 for no line

  lineBits.BIT2 = 0

  lineBits.BIT3 = 0

RETURN

Read_IR_Sensors:

  FREQOUT LfIrOut, 1, 38500                     ' modulate left IR LED

  irBits.BIT1 = ~LfIrIn                         ' read input (1 = target)

  FREQOUT RtIrOut, 1, 38500                     ' modulate right IR LED

  irBits.BIT0 = ~RtIrIn                         ' read input (1 = target)

  irBits.BIT2 = 0

  irBits.BIT3 = 0

  RETURN

' Movement

backup:

'  DEBUG "backup", CR

  FOR temp = 1 TO 50

    PULSOUT Lservo, (SStop - fast)

    PULSOUT Rservo, (SStop + fast)

    PAUSE 20

  NEXT

RETURN

forward:

'DEBUG "forwards", CR

  FOR temp = 0 TO 2

    PULSOUT Lservo, (SStop + fast)

    PULSOUT Rservo, (SStop - fast)

    PAUSE 20

  NEXT

RETURN

right:

'  DEBUG "right", CR

  PULSOUT LServo, (SStop + fast)

  PULSOUT RServo, (SStop + slow)

  PAUSE 20

RETURN

left:

'  DEBUG "left", CR

  PULSOUT LServo, (SStop - slow)

  PULSOUT RServo, (SStop - fast)

RETURN

' Debug variables

debugMsg:

  DEBUG CLS

  DEBUG ? leftsense, CR

  DEBUG ? rightSense, CR

  DEBUG "lineBits: ", BIN2 lineBits, CR

  DEBUG "irBits: ", BIN2 irBits, CR

RETURN

